Modeling interactions and the use of CONTRAST statement for post-fitting comparisons

Huiru Dong
Urban Health Research Initiative
British Columbia Centre for Excellence in HIV/AIDS
E-mail: hdong@cfenet.ubc.ca

BRITISH COLUMBIA CENTRE *for* EXCELLENCE *in* HIV/AIDS

Providence HEALTH CARE How you want to be treated.

UBC

Outline

- Concepts of effect-measure modification/ interaction
- SAS coding schemes
- Examples of post-fitting comparisons under different coding schemes
- Summary

Examples of effect-measure modification

Alcohol consumption

+

Driving

Injury



http://thewaytosobriety.com/drinking-and-driving-the-dangers-of-drunk-driving/

Examples of effect-measure modification

- Cigarette smoking during pregnancy is associated with low birth weight
- Maternal age is also associated with variations in birth weight
- Smoking has a bigger effect on risk of low birth weight in older moms than younger moms

http://brain-psychoneuron.blogspot.ca/2010/07/smoking-is-not-cooltime-for.html

Effect-measure modification

- Effect-measure modification refers to the situation in which a measure of effect changes over values of some other variable

Effect-measure modification

- Effect-measure modification refers to the situation in which a measure of effect changes over values of some other variable
  - Better understanding of causation
  - Identification of "high-risk" groups
  - Target interventions at specific subgroups

Measures of interaction

- Example:

  Additive statistical model

$$y = \beta_0 + \beta_1 x_1 + \beta_2 x_2 + \underline{\beta_3 (x_1 * x_2)} + \varepsilon$$

Measures of interaction

- Example:

Additive statistical model

$$y = \beta_0 + \beta_1 x_1 + \beta_2 x_2 + \qquad \varepsilon$$

Effect x_1 on y is measured by β_1

Measures of interaction

- Example:

Additive statistical model

$$y = \beta_0 + \beta_1 x_1 + \beta_2 x_2 + \beta_3 (x_1 * x_2) + \varepsilon$$

Effect x_1 on y is measured by $\beta_1 + \beta_3 x_2$

Outline

- Concepts of effect-measure modification/ interaction
- SAS coding schemes
- Examples of post-fitting comparisons under different coding schemes
- Summary

Hypothetical example

```
DATA Injury_data;
   INPUT Driving Alcohol Injury Count @@;
   DATALINES;
0   0   1  20     0   0  0  46
0   1   1  20     0   1  0  20
0   2   1  30     0   2  0  30
1   0   1  80     1   0  0  30
1   1   1  70     1   1  0  28
1   2   1  100    1   2  0  11
2   0   1  80     2   0  0  60
2   1   1  450    2   1  0  48
2   2   1  500    2   2  0  52
;
RUN;
```

Driving Skill:
Driving 0: Excellent
1: Good
2: Bad

Alcohol Consumption:
Alcohol 0: Low
1: Moderate
2: High

Outcome:
Injury 0: No
1: Yes

Coding Schemes

- Three coding schemes available in SAS:
  - Effect coding
  - Reference coding
  - Indicator (GLM or dummy) coding

Effect coding

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Ref=First;
   Model Injury= Driving Alcohol Driving*Alcohol;
RUN;
```

Effect coding

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Ref=First;
   Model Injury= Driving Alcohol Driving*Alcohol;
RUN;
```

- Default coding for LOGISTIC procedure
- Can be specified with the PARAM=EFFECT option in the CLASS statement for some other procedures (e.g. GENMOD)

Effect coding

Class Level Information

Class	Value	Design Variables	
		D_1	D_2
Driving	0	-1	-1
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	-1	-1
	1	1	0
	2	0	1

Replace the actual variable in the design matrix with a set of variables that use values of **-1** ,**0** or **1**

$$\log(Odds) = a + b_1 D_1 + b_2 D_2 + c_1 A_1 + c_2 A_2 + g_1 D_1 A_1 + g_2 D_1 A_2 + g_3 D_2 A_1 + g_4 D_2 A_2$$

Reference coding

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving(Ref='0') Alcohol(Ref='0') / Param=ref;
   Model Injury= Driving Alcohol Driving*Alcohol;
RUN;
```

Reference coding

Class Level Information

Class	Value	Design Variables	
		D_1	D_2
Driving	0	0	0
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	0	0
	1	1	0
	2	0	1

Replace the actual variable in the design matrix with a set of variables that use values of **0** or **1**

$$\log(Odds) = a + b_1D_1 + b_2D_2 + c_1A_1 + c_2A_2 + g_1D_1A_1 + g_2D_1A_2 + g_3D_2A_1 + g_4D_2A_2$$

Comparisons between effect and reference coding

- Effect coding
  - A main effect parameter is interpreted as the difference of the level's effect from the **average effect of all the levels**

- Reference coding
  - A main effect parameter is interpreted as the difference in the level's effect **compared to the reference level**

Comparisons between effect and reference coding

Effect coding

Parameter			DF	Estimate
Intercept			1	0.8956
Driving	1		1	0.4725
Driving	2		1	0.7007
Alcohol	1		1	0.1558
Alcohol	2		1	0.5946
Driving*Alcohol	1	1	1	-0.6077
Driving*Alcohol	1	2	1	0.2446
Driving*Alcohol	2	1	1	0.4859
Driving*Alcohol	2	2	1	0.0724

Reference coding

Parameter			DF	Estimate
Intercept			1	-0.8329
Driving	1		1	1.8137
Driving	2		1	1.1205
Alcohol	1		1	0.8329
Alcohol	2		1	0.8329
Driving*Alcohol	1	1	1	-0.8974
Driving*Alcohol	1	2	1	0.3936
Driving*Alcohol	2	1	1	1.1175
Driving*Alcohol	2	2	1	1.1428

Outline

- Concepts of effect-measure modification/ interaction
- SAS coding schemes
- Examples of post-fitting comparisons under different coding schemes
- Summary

Hypothetical example

- Question:

What's the odds ratio for having "bad" driving skill compared with "excellent" driving skill, when have "high" level of alcohol consumption before driving?

Driving Skill:

Driving	0:	Excellent
	1:	Good
	2:	Bad

Alcohol Consumption:

Alcohol	0:	Low
	1:	Moderate
	2:	High

Driving = 2 & Alcohol = 2

vs.

Driving = 0 & Alcohol = 2

Modeling interactions (Effect coding)

Class	Value	Design Variables	
		D_1	D_2
Driving	0	-1	-1
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	-1	-1
	1	1	0
	2	0	1

$$\log(Odds) = a + b_1D_1 + b_2D_2 + c_1A_1 + c_2A_2 + g_1D_1A_1 + g_2D_1A_2 + g_3D_2A_1 + g_4D_2A_2$$

Modeling interactions (Effect coding)

Class	Value	Design Variables	
		D_1	D_2
Driving	0	-1	-1
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	-1	-1
	1	1	0
	2	0	1

$$\log(Odds) = a + b_1 D_1 + b_2 D_2 + c_1 A_1 + c_2 A_2 + g_1 D_1 A_1 + g_2 D_1 A_2 + g_3 D_2 A_1 + g_4 D_2 A_2$$

Driving = 2 & Alcohol = 2

$$D_1 = 0, D_2 = 1$$
$$A_1 = 0, A_2 = 1$$

$$\log(Odds_{Driving2Alcohol2}) = a + b_2 + c_2 + g_4$$

Modeling interactions (Effect coding)

Class	Value	Design Variables	
		D_1	D_2
Driving	0	-1	-1
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	-1	-1
	1	1	0
	2	0	1

$$\log(Odds) = a + b_1D_1 + b_2D_2 + c_1A_1 + c_2A_2 + g_1D_1A_1 + g_2D_1A_2 + g_3D_2A_1 + g_4D_2A_2$$

Driving = 0 & Alcohol = 2

$$D_1 = -1, D_2 = -1$$
$$A_1 = 0, A_2 = 1$$

$$\log(Odds_{Driving0Alcohol2}) = a - b_1 - b_2 + c_2 - g_2 - g_4$$

Modeling interactions (Effect coding)

Driving = 2 & Alcohol = 2

vs.

Driving = 0 & Alcohol = 2

$$\log(Odds_{Driving2Alcohol2}) - \log(Odds_{Driving0Alcohol2}) = b_1 + 2b_2 + g_2 + 2g_4$$

$$Odds_{Driving2Alcohol2}/Odds_{Driving0Alcohol2} = exp(b_1 + 2b_2 + g_2 + 2g_4)$$

Modeling interactions (Effect coding)

Driving = 2 & Alcohol = 2 vs. Driving = 0 & Alcohol = 2

$$\log(Odds_{Driving2Alcohol2}) - \log(Odds_{Driving0Alcohol2}) = b_1 + 2b_2 + g_2 + 2g_4$$

$$Odds_{Driving2Alcohol2}/Odds_{Driving0Alcohol2} = exp(b_1 + 2b_2 + g_2 + 2g_4)$$

Driving: 1 2

$$\log(Odds) = a + b_1D_1 + b_2D_2 + c_1A_1 + c_2A_2 + g_1D_1A_1 + g_2D_1A_2 + g_3D_2A_1 + g_4D_2A_2$$

Driving*Alcohol: 0 1 0 2

Modeling interactions (Effect coding)

CONTRAST Statement:

CONTRAST ‘label’ row-description<,…,row-description></ options>;

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Ref=First;
   Model Injury= Driving Alcohol Driving*Alcohol;
   Contrast “D 2vs0 at A 2”
               Driving 1 2
               Driving*Alcohol 0 1 0 2 /Estimate=both;
RUN;
```

Modeling interactions (Effect coding)

Contrast Estimation and Testing Results by Row

Contrast	Type	Row	Estimate	Standard Error	Alpha	Confidence Limits		Wald Chi-Square	Pr > ChiSq
D 2vs0 at A 2	PARM	1	2.2634	0.2965	0.05	1.6823	2.8444	58.2818	<.0001
D 2vs0 at A 2	EXP	1	9.6154	2.8507	0.05	5.3778	17.1920	58.2818	<.0001

Modeling interactions (Reference coding)

Class	Value	Design Variables	
		D_1	D_2
Driving	0	0	0
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	0	0
	1	1	0
	2	0	1

$$\log(Odds) = a + b_1D_1 + b_2D_2 + c_1A_1 + c_2A_2 + g_1D_1A_1 + g_2D_1A_2 + g_3D_2A_1 + g_4D_2A_2$$

Modeling interactions (Reference coding)

Class	Value	Design Variables	
		D_1	D_2
Driving	0	0	0
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	0	0
	1	1	0
	2	0	1

$$\log(Odds) = a + b_1 D_1 + b_2 D_2 + c_1 A_1 + c_2 A_2 + g_1 D_1 A_1 + g_2 D_1 A_2 + g_3 D_2 A_1 + g_4 D_2 A_2$$

Driving = 2 & Alcohol = 2

$$D_1 = 0, D_2 = 1$$
$$A_1 = 0, A_2 = 1$$

$$\log(Odds_{Driving2Alcohol2}) = a + b_2 + c_2 + g_4$$

Modeling interactions (Reference coding)

Class	Value	Design Variables	
		D_1	D_2
Driving	0	0	0
	1	1	0
	2	0	1
		A_1	A_2
Alcohol	0	0	0
	1	1	0
	2	0	1

$$\log(Odds) = a + b_1 D_1 + b_2 D_2 + c_1 A_1 + c_2 A_2 + g_1 D_1 A_1 + g_2 D_1 A_2 + g_3 D_2 A_1 + g_4 D_2 A_2$$

Driving = 0 & Alcohol = 2

$$D_1 = 0, D_2 = 0$$
$$A_1 = 0, A_2 = 1$$

$$\log(Odds_{Driving0Alcohol2}) = a + c_2$$

BRITISH COLUMBIA CENTRE *for* EXCELLENCE *in* HIV/AIDS

Providence HEALTH CARE How you want to be treated.

UBC

Modeling interactions (Reference coding)

Driving = 2 & Alcohol = 2

vs.

Driving = 0 & Alcohol = 2

$$\log(Odds_{Driving2Alcohol2}) - \log(Odds_{Driving0Alcohol2}) = b_2 + g_4$$

$$Odds_{Driving2Alcohol2}/Odds_{Driving0Alcohol2} = exp(b_2 + g_4)$$

Modeling interactions (Reference coding)

Driving = 2 & Alcohol = 2

vs.

Driving = 0 & Alcohol = 2

$$\log(Odds_{Driving2Alcohol2}) - \log(Odds_{Driving0Alcohol2}) = b_2 + g_4$$

$$Odds_{Driving2Alcohol2}/Odds_{Driving0Alcohol2} = exp(b_2 + g_4)$$

Driving: **0** **1**

$$\log(Odds) = a + b_1D_1 + b_2D_2 + c_1A_1 + c_2A_2 + g_1D_1A_1 + g_2D_1A_2 + g_3D_2A_1 + g_4D_2A_2$$

Driving*Alcohol: **0** **0** **0** **1**

Modeling interactions (Reference coding)

CONTRAST Statement:

CONTRAST ‘label’ row-description<,…,row-description></ options>;

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving(Ref='0') Alcohol(Ref='0') / Param=ref;
   Model Injury= Driving Alcohol Driving*Alcohol;
   Contrast "D 2vs0 at A 2"
               Driving 0 1
               Driving*Alcohol 0 0 0 1 / Estimate=both;
RUN;
```

Modeling interactions (Reference coding)

Contrast Estimation and Testing Results by Row

Contrast	Type	Row	Estimate	Standard Error	Alpha	Confidence Limits		Wald Chi-Square	Pr > ChiSq
D 2vs0 at A 2	PARM	1	2.2634	0.2965	0.05	1.6823	2.8444	58.2818	<.0001
D 2vs0 at A 2	EXP	1	9.6154	2.8507	0.05	5.3778	17.1920	58.2818	<.0001

Summary

- Effect-measure modification is an important concept in public health research
- Understanding SAS coding scheme can help better interpret the parameter estimates
- Pay attention to the design matrix
- The selected parameterization method has a profound effect on how CONTRAST statements are specified and the associated hypothesis tests

References

- Rothman, K. J. (2012). *Epidemiology: an introduction*. Oxford University Press.
- Rothman, K. J., Greenland, S., & Lash, T. L. (Eds.). (2008). *Modern epidemiology*. Lippincott Williams & Wilkins.
- Kiernan, K. (2011). CONTRAST and ESTIMATE statements made easy: the LSMESTRIMATE statement. In *Statistics and Data Analysis SAS Global Forum*.
- Pritchard, M.L., Pasta, D.J. Head of the CLASS: Impress your colleagues with a superior understanding of the CLASS statement in PROC LOGISTIC. *Statistics and Data Analysis*.
- Pasta, D. J. (2011). Those confounded interactions: building and interpreting a model with many potential confounders and interactions. In *Statistics and Data Analysis SAS Global Forum*.
- Pasta, D. J. (2005). Parameterizing models to test the hypotheses you want: coding indicator variables and modified continuous variables. In *Proceedings of the Thirtieth Annual SAS Users Group International Conference* (pp. 212-30).
- Fox, S. H., Koepsell, T. D., & Daling, J. R. (1994). Birth weight and smoking during pregnancy-effect modification by maternal age. *American journal of epidemiology*, *139*(10), 1008-1015.

Additional information

Indicator (GLM or dummy) coding

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Param=glm;
   Model Injury= Driving Alcohol Driving*Alcohol;
RUN;
```

Indicator (GLM or dummy) coding

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Param=glm;
   Model Injury= Driving Alcohol Driving*Alcohol;
RUN;
```

- In PROC LOGISTIC, this can be specified with the PARAM=GLM option
- Default coding in procedures including GLM, MIXED, GLMMIX, GENMOD

Indicator (GLM or dummy) coding

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Param=glm;
   Model Injury= Driving Alcohol Driving*Alcohol;
RUN;
```

```
   Class Level Information

                              Design
Class        Value          Variables

Driving        0          1       0       0
               1          0       1       0
               2          0       0       1

Alcohol        0          1       0       0
               1          0       1       0
               2          0       0       1
```

Replace the actual variable in the design matrix with a set of variables that use values of 0 or 1

Modeling interactions (Indicator coding)

Class	Value	Design Variables		
		D_1	D_2	D_3
Driving	0	1	0	0
	1	0	1	0
	2	0	0	1
		A_1	A_2	A_3
Alcohol	0	1	0	0
	1	0	1	0
	2	0	0	1

$$\log(Odds) = a + b_1D_1 + b_2D_2 + b_3D_3$$
$$+c_1A_1 + c_2A_2 + c_3A_3$$
$$+ g_1D_1A_1 + g_2D_1A_2 + g_3D_1A_3$$
$$+ g_4D_2A_1 + g_5D_2A_2 + g_6D_2A_3$$
$$+ g_7D_3A_1 + g_8D_3A_2 + g_9D_3A_3$$

Modeling interactions (Indicator coding)

Class	Value	Design Variables		
		D_1	D_2	D_3
Driving	0	1	0	0
	1	0	1	0
	2	0	0	1
		A_1	A_2	A_3
Alcohol	0	1	0	0
	1	0	1	0
	2	0	0	1

$$\log(Odds) = a + b_1D_1 + b_2D_2 + b_3D_3$$
$$+c_1A_1 + c_2A_2 + c_3A_3$$
$$+ g_1D_1A_1 + g_2D_1A_2 + g_3D_1A_3$$
$$+ g_4D_2A_1 + g_5D_2A_2 + g_6D_2A_3$$
$$+ g_7D_3A_1 + g_8D_3A_2 + g_9D_3A_3$$

Driving = 2 & Alcohol = 2

$$D_1 = 0, D_2 = 0, D_3 = 1$$
$$A_1 = 0, A_2 = 0, A_3 = 1$$

$$\log(Odds_{Driving2Alcohol2}) = a + b_3 + c_3 + g_9$$

Modeling interactions (Indicator coding)

Class	Value	Design Variables		
		D_1	D_2	D_3
Driving	0	1	0	0
	1	0	1	0
	2	0	0	1
		A_1	A_2	A_3
Alcohol	0	1	0	0
	1	0	1	0
	2	0	0	1

$$\log(Odds) = a + b_1D_1 + b_2D_2 + b_3D_3$$
$$+c_1A_1 + c_2A_2 + c_3A_3$$
$$+ g_1D_1A_1 + g_2D_1A_2 + g_3D_1A_3$$
$$+ g_4D_2A_1 + g_5D_2A_2 + g_6D_2A_3$$
$$+ g_7D_3A_1 + g_8D_3A_2 + g_9D_3A_3$$

Driving = 0 & Alcohol = 2

$$D_1 = 1, D_2 = 0, D_3 = 0$$
$$A_1 = 0, A_2 = 0, A_3 = 1$$

$$\log(Odds_{Driving0Alcohol2}) = a + b_1 + c_3 + g_3$$

Modeling interactions (Indicator coding)

Driving = 2 & Alcohol = 2

vs.

Driving = 0 & Alcohol = 2

$$\log(Odds_{Driving2Alcohol2}) - \log(Odds_{Driving0Alcohol2}) = -b_1 + b_3 - g_3 + g_9$$

$$Odds_{Driving2Alcohol2} / Odds_{Driving0Alcohol2} = exp(-b_1 + b_3 - g_3 + g_9)$$

$$\log(Odds) = a + b_1 D_1 + b_2 D_2 + b_3 D_3 + c_1 A_1 + c_2 A_2 + c_3 A_3 + g_1 D_1 A_1 + g_2 D_1 A_2 + g_3 D_1 A_3 + g_4 D_2 A_1 + g_5 D_2 A_2 + g_6 D_2 A_3 + g_7 D_3 A_1 + g_8 D_3 A_2 + g_9 D_3 A_3$$

Modeling interactions (Indicator coding)

Driving = 2 & Alcohol = 2

vs.

Driving = 0 & Alcohol = 2

$$\log(Odds_{Driving2Alcohol2}) - \log(Odds_{Driving0Alcohol2}) = -b_1 + b_3 - g_3 + g_9$$

$$Odds_{Driving2Alcohol2}/Odds_{Driving0Alcohol2} = exp(-b_1 + b_3 - g_3 + g_9)$$

$$\log(Odds) = a + \overset{-1}{b_1D_1} + \overset{0}{b_2D_2} + \overset{1}{b_3D_3} + c_1A_1 + c_2A_2 + c_3A_3$$
$$+ \underset{0}{g_1D_1A_1} + \underset{0}{g_2D_1A_2} + \underset{-1}{g_3D_1A_3} + \underset{0}{g_4D_2A_1} + \underset{0}{g_5D_2A_2} + \underset{0}{g_6D_2A_3}$$
$$+ \underset{0}{g_7D_3A_1} + \underset{0}{g_8D_3A_2} + \underset{1}{g_9D_3A_3}$$

Modeling interactions (Indicator coding)

CONTRAST Statement:

CONTRAST ‘label’ row-description<,…,row-description></ options>;

```
PROC LOGISTIC Data=Injury_data Descending;
   Freq  count;
   Class Driving Alcohol / Param=glm;
   Model Injury= Driving Alcohol Driving*Alcohol;
   Contrast "D 2vs0 at A 2"
             Driving -1 0 1
             Driving*Alcohol 0 0 -1 0 0 0 0 0 1 /E Estimate=both;
RUN;
```

Modeling interactions (Indicator coding)

Coefficients of Contrast D 2vs0 at A 2

Parameter	Row1
Intercept	0
Driving0	-1
Driving1	0
Driving2	1
Alcohol0	0
Alcohol1	0
Alcohol2	0
Driving0Alcohol0	0
Driving0Alcohol1	0
Driving0Alcohol2	-1
Driving1Alcohol0	0
Driving1Alcohol1	0
Driving1Alcohol2	0
Driving2Alcohol0	0
Driving2Alcohol1	0
Driving2Alcohol2	1

Modeling interactions (Indicator coding)

Contrast Estimation and Testing Results by Row

Contrast	Type	Row	Estimate	Standard Error	Alpha	Confidence Limits		Wald Chi-Square	Pr > ChiSq
D 2vs0 at A 2	PARM	1	2.2634	0.2965	0.05	1.6823	2.8444	58.2818	<.0001
D 2vs0 at A 2	EXP	1	9.6154	2.8507	0.05	5.3778	17.1920	58.2818	<.0001

Modeling interactions (Effect coding)

Analysis of Maximum Likelihood Estimates

Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		1	0.8956	0.0792	127.7505	<.0001
Driving	1	1	0.4725	0.1165	16.4577	<.0001
Driving	2	1	0.7007	0.0949	54.5624	<.0001
Alcohol	1	1	0.1558	0.1126	1.9135	0.1666
Alcohol	2	1	0.5946	0.1152	26.6379	<.0001
Driving*Alcohol	1 1	1	-0.6077	0.1598	14.4630	0.0001
Driving*Alcohol	1 2	1	0.2446	0.1782	1.8825	0.1701
Driving*Alcohol	2 1	1	0.4859	0.1341	13.1360	0.0003
Driving*Alcohol	2 2	1	0.0724	0.1355	0.2857	0.5930

Modeling interactions (Reference coding)

Analysis of Maximum Likelihood Estimates

Parameter			DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept			1	-0.8329	0.2678	9.6692	0.0019
Driving	1		1	1.8137	0.3429	27.9784	<.0001
Driving	2		1	1.1205	0.3177	12.4435	0.0004
Alcohol	1		1	0.8329	0.4144	4.0390	0.0445
Alcohol	2		1	0.8329	0.3720	5.0117	0.0252
Driving*Alcohol	1	1	1	-0.8974	0.5173	3.0097	0.0828
Driving*Alcohol	1	2	1	0.3936	0.5340	0.5433	0.4611
Driving*Alcohol	2	1	1	1.1175	0.4732	5.5762	0.0182
Driving*Alcohol	2	2	1	1.1428	0.4345	6.9176	0.0085

Modeling interactions (Indicator coding)

Analysis of Maximum Likelihood Estimates

Parameter			DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept			1	2.2634	0.1457	241.2922	<.0001
Driving	0		1	-2.2634	0.2965	58.2818	<.0001
Driving	1		1	-0.0561	0.3495	0.0258	0.8725
Driving	2		0	0	.	.	.
Alcohol	0		1	-1.9757	0.2245	77.4508	<.0001
Alcohol	1		1	-0.0253	0.2104	0.0145	0.9042
Alcohol	2		0	0	.	.	.
Driving*Alcohol	0	0	1	1.1428	0.4345	6.9176	0.0085
Driving*Alcohol	0	1	1	0.0253	0.4593	0.0030	0.9560
Driving*Alcohol	0	2	0	0	.	.	.
Driving*Alcohol	1	0	1	0.7492	0.4440	2.8475	0.0915
Driving*Alcohol	1	1	1	-1.2657	0.4418	8.2067	0.0042
Driving*Alcohol	1	2	0	0	.	.	.
Driving*Alcohol	2	0	0	0	.	.	.
Driving*Alcohol	2	1	0	0	.	.	.
Driving*Alcohol	2	2	0	0	.	.	.